
    HtmlQuery

A concise API, honed over multiple years, for querying HTML. There are just 5 main functions:
all/2, find/2 and find!/2 for finding things, plus attr/2 and text/1 for extracting
information. There are also a handful of other useful functions, referenced below and described in detail in
the module docs. HTML parsing is handled by
Floki.
The input can be:
	A string of HTML.
	An IO Data of HTML.
	A Floki html_node
or html_tree data structure. HtmlQuery uses Floki internally
and can accept its data structure as input, and some HtmlQuery functions return its data structure as output.
	Anything that implements the String.Chars protocol. See Implementing String.Chars
below.

We created a related library called XmlQuery which has the same API but
is used for querying XML. You can read more about them in
Querying HTML and XML in Elixir with HtmlQuery and XmlQuery.
This library is MIT licensed and is part of a growing number of Elixir open source libraries published at
github.com/synchronal.
This library is tested against the most recent 3 versions of Elixir and Erlang.
Sponsorship 💕
This library is part of the Synchronal suite of libraries and tools
which includes more than 15 open source Elixir libraries as well as some Rust libraries and tools.
You can support our open source work by sponsoring us.
If you have specific features in mind, bugs you'd like fixed, or new libraries you'd like to see,
file an issue or contact us at contact@reflective.dev.
Installation
def deps do
  [
    {:html_query, "~> 4.4"}
  ]
end
Usage
Detailed docs are in the HtmlQuery module docs; a quick usage
overview follows.
We typically alias HtmlQuery to Hq:
alias HtmlQuery, as: Hq
The rest of these examples use the following HTML:
html = """
  <h1>Please update your profile</h1>
  <form id="profile" test-role="profile">
    <label>Name <input name="name" type="text" value="Fido"> </label>
    <label>Age <input name="age" type="text" value="10"> </label>
    <label>Bio <textarea name="bio">Fido likes long walks and playing fetch.</textarea> </label>
  </form>
</form>
"""
Querying
Query functions use CSS selector strings for finding nodes. The
MDN CSS Selectors guide
is a helpful CSS reference.
Hq.find(html, "form#profile label textarea")
We’ve found that reserving CSS classes and IDs for styling instead of using them for testing reduces the chance of
styling changes breaking tests, and so we often add attributes that start with test- into our HTML; a query for
test-role would look like:
Hq.find(html, "form[test-role=profile]")
For simple queries, HtmlQuery.Css provides a shorthand
using keyword lists. For complicated queries, it’s usually clearer to use a CSS string.
Hq.find(html, test_role: "profile")
Finding
all/2 finds all elements matching the query, find/2 returns the first element that matches the selector or nil if
none was found, and find!/2 is like find/2 but raises unless exactly one element is found.
Hq.all(html, "input") # returns a list of all the <input> elements
Hq.find(html, "input[name=age]") # returns the <input> with `name=age`
Hq.find!(html, "input[name=foo]") # raises because no such element exists
See the module docs for more details.
Extracting
text/1 is the simplest extraction function:
html |> Hq.find(:h1) |> Hq.text() # returns "Please update your profile"
attr/2 returns the value of an attribute:
html |> Hq.find("input[name=age]") |> Hq.attr(:value) # returns "10"
To extract data from multiple HTML nodes, we found that it is clearer to compose multiple functions rather than to
have a more complicated API:
html |> Hq.all(:input) |> Enum.map(&Hq.text/1) # returns ["Name", "Age"]
html |> Hq.all("input[type=text]") |> Enum.map(&Hq.attr(&1, "value")) # returns ["Fido", "10"]
There are also functions for extracting form fields as a map, meta tags as a list, and table contents as a list of
lists or a list of maps. See the module docs for more details.
Parsing
parse/1 and parse_doc/1 delegate to Floki’s parse_fragment/1 and parse_document!/1 functions. These functions
are rarely needed since all the HtmlQuery functions will parse HTML if needed. See the
module docs for more details.
Utilities
inspect_html/2 prints prettified HTML with a label, normalize/1 parses and re-stringifies HTML which can be handy
when trying to compare two strings of HTML, pretty/1 formats HTML in a human-friendly format, and reject/2 removes
nodes that match the given selector. See the module docs for more
details.
Implementing String.Chars
HtmlQuery functions that accept HTML will convert any module that implements String.Chars. For example, our
Pages testing library implements String.Chars for controller output like
this:
defimpl String.Chars do
  def to_string(%Pages.Driver.Conn{conn: %Plug.Conn{status: 200} = conn}),
    do: Phoenix.ConnTest.html_response(conn, 200)
end
and implements String.Chars for LiveView output like this:
defimpl String.Chars, for: Pages.Driver.LiveView do
  def to_string(%Pages.Driver.LiveView{rendered: rendered}) when not is_nil(rendered),
    do: rendered

  def to_string(%Pages.Driver.LiveView{live: live}) when not is_nil(live),
    do: live |> Phoenix.LiveViewTest.render()
end
Development
brew bundle

bin/dev/doctor
bin/dev/update
bin/dev/audit
bin/dev/shipit



  

    LICENSE

Copyright 2022 synchronal.dev
Permission is hereby granted, free of charge, to any person obtaining a copy of this software and associated documentation files (the "Software"), to deal in the Software without restriction, including without limitation the rights to use, copy, modify, merge, publish, distribute, sublicense, and/or sell copies of the Software, and to permit persons to whom the Software is furnished to do so, subject to the following conditions:
The above copyright notice and this permission notice shall be included in all copies or substantial portions of the Software.
THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE SOFTWARE.

A concise HTML query API. HTML parsing is handled by Floki.
We created a related library called XmlQuery which has the same API but
is used for querying XML. You can read more about them in
Querying HTML and XML in Elixir with HtmlQuery and XmlQuery.
Data types
All functions can accept HTML in the form of a string, a Floki HTML tree, a Floki HTML node, or anything that
implements the String.Chars protocol. See HtmlQuery.html/0.
Some functions take a CSS selector, which can be a string, a keyword list, or a list.
See HtmlQuery.Css.selector/0.
Query functions
	all/2	return all elements matching the selector
	find/2	return the first element that matches the selector
	find!/2	return the only element that matches the selector, or raise

Extraction functions
	attr/2	returns the attribute value as a string
	form_fields/1	returns the names and values of form fields as a map
	meta_tags/1	returns the names and values of metadata fields
	table/2	returns the cells of a table as a list of lists or maps
	text/2	returns the text contents as a single string

Parsing functions
	parse/1	parses an HTML fragment into a [Floki HTML tree]
	parse_doc/1	parses an HTML doc into a [Floki HTML tree]

Utility functions
	inspect_html/2	prints prettified HTML with a label
	normalize/1	parses and re-stringifies HTML
	pretty/1	prettifies HTML
	reject/2	removes nodes that match the selector

Alias
If you use HtmlQuery a lot, you may want to alias it to the recommended shortcut "Hq":
alias HtmlQuery, as: Hq
Examples
Get the value of a selected option:
iex> html = ~s|<select> <option value="a" selected>apples</option> <option value="b">bananas</option> </select>|
iex> HtmlQuery.find(html, "select option[selected]") |> HtmlQuery.attr("value")
"a"
Get the text of a selected option, raising if there are more than one:
iex> html = ~s|<select> <option value="a" selected>apples</option> <option value="b">bananas</option> </select>|
iex> HtmlQuery.find!(html, "select option[selected]") |> HtmlQuery.text()
"apples"
Get the text of all the options:
iex> html = ~s|<select> <option value="a" selected>apples</option> <option value="b">bananas</option> </select>|
iex> HtmlQuery.all(html, "select option") |> Enum.map(&HtmlQuery.text/1)
["apples", "bananas"]
Use a keyword list as the selector (see HtmlQuery.Css for details on selectors):
iex> html = ~s|<div> <a href="/logout" test-role="logout-link">logout</a> </div>|
iex> HtmlQuery.find!(html, test_role: "logout-link") |> HtmlQuery.attr("href")
"/logout"

      


      
        Summary


  
    Types
  


    
      
        attr()

      


        A string or atom representing an attribute name. If an atom, underscores are converted to dashes.



    


    
      
        html()

      


        A string, a struct that implements the String.Chars protocol,
a Floki HTML tree, or a Floki HTML node.



    





  
    Functions
  


    
      
        all(html, selector)

      


        Finds all elements in html that match selector, returning a Floki HTML tree.



    


    
      
        attr(html, attr)

      


        Returns the value of attr from the outermost element of html.
If attr is an atom, any underscores are converted to dashes.



    


    
      
        find(html, selector)

      


        Finds the first element in html that matches selector, returning a Floki HTML node.



    


    
      
        find!(html, selector)

      


        Like find/2 but raises unless exactly one element is found.



    


    
      
        form(html, opts \\ [])

      


        The beginnings of an experimental replacement for HtmlQuery.form_fields/1.



    


    
      
        form_fields(html)

      


        Returns a map containing the form fields of form selector in html. Because it returns a map, any information
about the order of form fields is lost.



    


    
      
        inspect_html(html, label \\ "INSPECTED HTML")

      


        Prints prettified html with a label, and then returns the original html.



    


    
      
        meta_tags(html)

      


        Extracts all the meta tags from html, returning a list of maps.



    


    
      
        normalize(html)

      


        Parses and then re-stringifies html, increasing the liklihood that two equivalent HTML strings can
be considered equal.



    


    
      
        parse(html)

      


        Parses an HTML fragment using Floki.parse_fragment!/1, returning a Floki HTML tree.



    


    
      
        parse_doc(html)

      


        Parses an HTML document using Floki.parse_document!/1, returning a Floki HTML tree.



    


    
      
        pretty(html)

      


        Returns html as a prettified string (delgates to Floki.raw_html/2 with the pretty: true option).



    


    
      
        reject(html, selector)

      


        Returns html after removing all nodes that don't match selector (delegates to Floki.filter_out/2).



    


    
      
        table(html, opts \\ [])

      


        Returns the contents of the table as a list of lists, or as a single list if only one column is selected via
the :only or :except options. Headers are returned by default (see the :headers option).



    


    
      
        text(html, separator \\ " ")

      


        Returns the text value of html, separating substrings with a space by default. (Floki will split text into
substrings.) You can pass a separator as the second argument; sometimes it's useful to pass an empty string.

      attr()



        
          
        

    

  


  

      

          @type attr() :: binary() | atom()


      


A string or atom representing an attribute name. If an atom, underscores are converted to dashes.

  



  
    
      
    
    
      html()



        
          
        

    

  


  

      

          @type html() :: binary() | String.Chars.t() | Floki.html_tree() | Floki.html_node()


      


A string, a struct that implements the String.Chars protocol,
a Floki HTML tree, or a Floki HTML node.

  


        

      

      
        Functions

        


  
    
      
    
    
      all(html, selector)



        
          
        

    

  


  

      

          @spec all(html(), HtmlQuery.Css.selector()) :: Floki.html_tree()


      


Finds all elements in html that match selector, returning a Floki HTML tree.
iex> html = ~s|<select> <option value="a" selected>apples</option> <option value="b">bananas</option> </select>|
iex> HtmlQuery.all(html, "option")
[
  {"option", [{"value", "a"}, {"selected", "selected"}], ["apples"]},
  {"option", [{"value", "b"}], ["bananas"]}
]

  



  
    
      
    
    
      attr(html, attr)



        
          
        

    

  


  

      

          @spec attr(html(), attr()) :: binary() | nil


      


Returns the value of attr from the outermost element of html.
If attr is an atom, any underscores are converted to dashes.
iex> html = ~s|<div> <a href="/logout" test-role="logout-link">logout</a> </div>|
iex> HtmlQuery.find!(html, test_role: "logout-link") |> HtmlQuery.attr("href")
"/logout"

  



  
    
      
    
    
      find(html, selector)



        
          
        

    

  


  

      

          @spec find(html(), HtmlQuery.Css.selector()) :: Floki.html_node() | nil


      


Finds the first element in html that matches selector, returning a Floki HTML node.
iex> html = ~s|<select> <option value="a" selected>apples</option> <option value="b">bananas</option> </select>|
iex> HtmlQuery.find(html, "select option[selected]")
{"option", [{"value", "a"}, {"selected", "selected"}], ["apples"]}

  



  
    
      
    
    
      find!(html, selector)



        
          
        

    

  


  

      

          @spec find!(html(), HtmlQuery.Css.selector()) :: Floki.html_node()


      


Like find/2 but raises unless exactly one element is found.

  



    

  
    
      
    
    
      form(html, opts \\ [])



        
          
        

    

  


  

      

          @spec form(html(), Keyword.t()) :: [Keyword.t()]


      


The beginnings of an experimental replacement for HtmlQuery.form_fields/1.
Returns all the fields of the supplied form, in order, without trying to emulate what a web framework might
convert the values to.
The :as option can be :lists or :map. When :lists (the default), each input's attributes are returned as
a list in the order found in the HTML, and therefore the function returns a list of lists. Because HTML allows
multiple inputs with the same name, no de-duplication is performed. When :map, the fields are returned as a
map, where the keys are the "name" attribute (converted into snake-cased atoms) and the values are a list of
lists of attributes (to handle the case when there are multiple inputs with the same name).

  



  
    
      
    
    
      form_fields(html)



        
          
        

    

  


  

      

          @spec form_fields(html()) :: %{required(atom()) => binary() | map()}


      


Returns a map containing the form fields of form selector in html. Because it returns a map, any information
about the order of form fields is lost.
iex> html = ~s|<form> <input type="text" name="color" value="green"> <textarea name="desc">A tree</textarea> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{color: "green", desc: "A tree"}
Field names are converted to snake case atoms:
iex> html = ~s|<form> <input type="text" name="favorite-color" value="green"> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{favorite_color: "green"}
If form field names are in foo[bar] format, then foo becomes a key to a nested map containing bar:
iex> html = ~s|<form> <input type="text" name="profile[name]" value="fido"> <input type="text" name="profile[age]" value="10"> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{profile: %{name: "fido", age: "10"}}
If a text field has no value attribute, it will not be returned at all:
iex> html = ~s|<form> <input type="text" name="no-value"> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{}

iex> html = ~s|<form> <input type="text" name="empty-value" value=""> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{empty_value: ""}

iex> html = ~s|<form> <input type="text" name="non-empty-value" value="something"> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{non_empty_value: "something"}
The checked value of a radio button set is returned, or nil is returned if no value is checked:
iex> html = ~s|<form> <input type="radio" name="x" value="1"> <input type="radio" name="x" value="2" checked> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{x: "2"}

iex> html = ~s|<form> <input type="radio" name="x" value="1"> <input type="radio" name="x" value="2"> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{x: nil}
When evaluating checkboxes, the name attribute of the input defines whether or not a term or a
list will be returned. A name that ends in [] allows a browser to send multiple values, in which case
our form fields will return an array of values. A name that does not end in [] will evaluate to a
single value, the last checked value in a list:
iex> html = ~s|<form> <input type="checkbox" name="x" value="1" checked> <input type="checkbox" name="x" value="2" checked> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{x: "2"}

iex> html = ~s|<form> <input type="checkbox" name="x" value="1"> <input type="checkbox" name="x" value="2"> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{x: nil}

iex> html = ~s|<form>
...>   <input type="hidden" name="x" value="false">
...>   <input type="checkbox" name="x" value="true">
...>   <input type="hidden" name="y" value="false">
...>   <input type="checkbox" name="y" checked value="true">
...> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{x: false, y: true}

iex> html = ~s|<form> <input type="checkbox" name="x[]" value="1" checked> <input type="checkbox" name="x[]" value="2" checked> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{x: ["1", "2"]}

iex> html = ~s|<form> <input type="checkbox" name="x[]" value="1"> <input type="checkbox" name="x[]" value="2"> </form>|
iex> html |> HtmlQuery.find("form") |> HtmlQuery.form_fields()
%{x: []}

  



    

  
    
      
    
    
      inspect_html(html, label \\ "INSPECTED HTML")



        
          
        

    

  


  

      

          @spec inspect_html(html(), binary()) :: html()


      


Prints prettified html with a label, and then returns the original html.

  



  
    
      
    
    
      meta_tags(html)



        
          
        

    

  


  

      

          @spec meta_tags(html()) :: [%{required(binary()) => binary()}]


      


Extracts all the meta tags from html, returning a list of maps.
iex> html = ~s|<head> <meta charset="utf-8"/> <meta http-equiv="X-UA-Compatible" content="IE=edge"/> </head>|
iex> HtmlQuery.meta_tags(html)
[%{"charset" => "utf-8"}, %{"content" => "IE=edge", "http-equiv" => "X-UA-Compatible"}]

  



  
    
      
    
    
      normalize(html)



        
          
        

    

  


  

      

          @spec normalize(html()) :: binary()


      


Parses and then re-stringifies html, increasing the liklihood that two equivalent HTML strings can
be considered equal.
iex> a = ~s|<p id="color">green</p>|
iex> b = ~s|<p  id = "color" >green</p>|
iex> a == b
false
iex> HtmlQuery.normalize(a) == HtmlQuery.normalize(b)
true

  



  
    
      
    
    
      parse(html)



        
          
        

    

  


  

      

          @spec parse(html()) :: Floki.html_tree()


      


Parses an HTML fragment using Floki.parse_fragment!/1, returning a Floki HTML tree.

  



  
    
      
    
    
      parse_doc(html)



        
          
        

    

  


  

      

          @spec parse_doc(html()) :: Floki.html_tree()


      


Parses an HTML document using Floki.parse_document!/1, returning a Floki HTML tree.

  



  
    
      
    
    
      pretty(html)



        
          
        

    

  


  

      

          @spec pretty(html()) :: binary()


      


Returns html as a prettified string (delgates to Floki.raw_html/2 with the pretty: true option).

  



  
    
      
    
    
      reject(html, selector)



        
          
        

    

  


  

      

          @spec reject(html(), HtmlQuery.Css.selector()) :: html()


      


Returns html after removing all nodes that don't match selector (delegates to Floki.filter_out/2).
iex> html = ~s|<div> <span id="name">Alice</span> <span id="password">topaz</span> </div>|
iex> HtmlQuery.reject(html, id: "password") |> HtmlQuery.normalize()
~s|<div><span id="name">Alice</span></div>|

  



    

  
    
      
    
    
      table(html, opts \\ [])



        
          
        

    

  


  

      

          @spec table(
  html(),
  keyword()
) :: [[]] | [map()]


      


Returns the contents of the table as a list of lists, or as a single list if only one column is selected via
the :only or :except options. Headers are returned by default (see the :headers option).
Options:
	:as	if :lists (the default), returns the table as a list of lists; or
	if :maps, returns the table as a list of maps.


	:only	a list of the indices of the columns to return; or
	a single index of the column, which will return a single list, rather than a list of lists; or
	a list of column header names to return, assuming that the first row of the table is the column names; or
	a single column header name, which will return a single list, rather than a list of lists; or
	:all to return all columns (which is the same as not specifying this option at all)


	:except	returns all the columns except the ones whose indices or names are given. only and except can be
combined to further reduce the set of columns


	:headers	if true (the default), returns the list of headers along with the rows. Ignored if as option is :maps


	:update	accepts a function that will be used to transform each header and body cell



Deprecated options:
	:columns - use only instead.

iex> html = "<table> <tr><th>A</th><th>B</th><th>C</th></tr> <tr><td>1</td><td>2</td><td>3</td></tr> </table>"
iex> HtmlQuery.table(html)
[
  ["A", "B", "C"],
  ["1", "2", "3"]
]
iex> HtmlQuery.table(html, as: :maps)
[
  %{"A" => "1", "B" => "2", "C" => "3"}
]
iex> HtmlQuery.table(html, only: [0, 2])
[
  ["A", "C"],
  ["1", "3"]
]
iex> HtmlQuery.table(html, only: [2, 0])
[
  ["C", "A"],
  ["3", "1"]
]
iex> HtmlQuery.table(html, only: 2)
["C", "3"]
iex> HtmlQuery.table(html, only: ["C", "A"])
[
  ["C", "A"],
  ["3", "1"]
]
iex> HtmlQuery.table(html, only: "C")
["C", "3"]
iex> HtmlQuery.table(html, only: "C", headers: false)
["3"]
iex> HtmlQuery.table(html, except: ["C", "A"])
[
  ["B"],
  ["2"]
]
iex> HtmlQuery.table(html, only: ["C", "A"], headers: false)
[
  ["3", "1"]
]
iex> HtmlQuery.table(html, update: &String.downcase/1)
[
  ["a", "b", "c"],
  ["1", "2", "3"]
]

  



    

  
    
      
    
    
      text(html, separator \\ " ")



        
          
        

    

  


  

      

          @spec text(html(), String.t()) :: binary()


      


Returns the text value of html, separating substrings with a space by default. (Floki will split text into
substrings.) You can pass a separator as the second argument; sometimes it's useful to pass an empty string.
iex> html = ~s|<select> <option value="a" selected>apples</option> <option value="b">bananas</option> </select>|
iex> HtmlQuery.find!(html, "select option[selected]") |> HtmlQuery.text()
"apples"

Constructs CSS selectors via Elixir data structures. See selector/1 for details.

      


      
        Summary


  
    Types
  


    
      
        selector()

      


        See docs for selector/1 for more details.



    





  
    Functions
  


    
      
        selector(input)

      


        Accepts a string, atom, keyword list, or list, and returns a CSS string.



    





      


      
        Types

        


  
    
      
    
    
      selector()



        
          
        

    

  


  

      

          @type selector() :: binary() | atom() | keyword() | list()


      


See docs for selector/1 for more details.

  


        

      

      
        Functions

        


  
    
      
    
    
      selector(input)



        
          
        

    

  


  

      

          @spec selector(selector()) :: binary()


      


Accepts a string, atom, keyword list, or list, and returns a CSS string.
String syntax
When given a string, returns the string. This is useful when you don't know if a selector is already a string.
iex> HtmlQuery.Css.selector(".profile[test-role='new-members']")
".profile[test-role='new-members']"
Atom syntax
When given an atom, converts the atom to a string, without converting underscores to dashes.
iex> HtmlQuery.Css.selector(:p)
"p"
Keyword list syntax
The keyword list syntax is intentionally limited; complex selectors are more
easily written as strings.
The keyword list syntax makes it a bit eaiser to write simple selectors or selectors that use variables:
e.g.:
HtmlQuery.Css.selector(test_role: "new-members")
HtmlQuery.Css.selector(id: some_variable)
Keys are expected to be atoms and will be dasherized (foo_bar -> foo-bar).
Values are expected to be strings or another keyword list.
A key/value pair will be converted to an attribute selector:
iex> HtmlQuery.Css.selector(test_role: "new-members")
"[test-role='new-members']"
A keyword list will be converted to a list of attribute selectors:
iex> HtmlQuery.Css.selector(class: "profile", test_role: "new-members")
"[class='profile'][test-role='new-members']"
(Note that the CSS selector .profile expands to [class~='profile'] which is not equivalent to
[class='profile']. The keyword list syntax will not generate ~= selectors so you should use a
string selector such as ".profile[test-role='new-members']" instead if you want ~= semantics. See the
CSS spec for details.)
When the value is a keyword list, the key is converted to an element selector:
iex> HtmlQuery.Css.selector(p: [class: "profile", test_role: "new-members"])
"p[class='profile'][test-role='new-members']"
List syntax
The list syntax is intentionally limited; complex selectors are more
easily written as strings.
When the value is a list (vs a keyword list), atoms are converted to element selectors and
keyword lists are converted as described above.
iex> HtmlQuery.Css.selector([[p: [class: "profile", test_role: "new-members"]], :div, [class: "tag"]])
"p[class='profile'][test-role='new-members'] div [class='tag']"

  


        

      


  

    
HtmlQuery.Form 
    



      
Form-handling functions. Most of the time, HtmlQuery.form_fields/1 should be used instead of calling these
functions directly.

HtmlQuery.QueryError exception
    



      
An exception raised when unable to find an HTML element.

